

NAG Library Routine Document

F06WCF (DSFRK)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F06WCF (DSFRK) performs one of the symmetric rank- k update operations

$$C \leftarrow \alpha AA^T + \beta C \quad \text{or} \quad C \leftarrow \alpha A^T A + \beta C,$$

where A is a real matrix, C is an n by n real symmetric matrix stored in Rectangular Full Packed (RFP) format, and α and β are real scalars.

2 Specification

```
SUBROUTINE F06WCF (TRANSR, UPLO, TRANS, N, K, ALPHA, A, LDA, BETA, CR)
INTEGER          N, K, LDA
REAL (KIND=nag_wp) ALPHA, A(LDA,*), BETA, CR(N*(N+1)/2)
CHARACTER(1)    TRANSR, UPLO, TRANS
```

The routine may be called by its LAPACK name *dsfrk*.

3 Description

F06WCF (DSFRK) performs one of the symmetric rank- k update operations

$$C \leftarrow \alpha AA^T + \beta C \quad \text{or} \quad C \leftarrow \alpha A^T A + \beta C,$$

where A is a real matrix, C is an n by n real symmetric matrix stored in Rectangular Full Packed (RFP) format, and α and β are real scalars. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction.

If $n = 0$ or if $\beta = 1.0$ and either $k = 0$ or $\alpha = 0.0$ then F06WCF (DSFRK) returns immediately. If $\beta = 0.0$ and either $k = 0$ or $\alpha = 0.0$ then C is set to the zero matrix.

4 References

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37**, 2

5 Parameters

1: TRANSR – CHARACTER(1) *Input*

On entry: specifies whether the RFP representation of C is normal or transposed.

TRANSR = 'N'

The matrix C is stored in normal RFP format.

TRANSR = 'T'

The matrix C is stored in transposed RFP format.

Constraint: TRANSR = 'N' or 'T'.

- 2: UPLO – CHARACTER(1) *Input*
On entry: specifies whether the upper or lower triangular part of C is stored in RFP format.
 UPLO = 'U'
 The upper triangular part of C is stored in RFP format.
 UPLO = 'L'
 The lower triangular part of C is stored in RFP format.
Constraint: UPLO = 'U' or 'L'.
- 3: TRANS – CHARACTER(1) *Input*
On entry: specifies the operation to be performed.
 TRANS = 'N'
 $C \leftarrow \alpha AA^T + \beta C$.
 TRANS = 'T'
 $C \leftarrow \alpha A^T A + \beta C$.
Constraint: TRANS = 'N' or 'T'.
- 4: N – INTEGER *Input*
On entry: n , the order of the matrix C .
Constraint: $N \geq 0$.
- 5: K – INTEGER *Input*
On entry: k , the number of columns of A if TRANS = 'N', or the number of rows of A if TRANS = 'T'.
Constraint: $K \geq 0$.
- 6: ALPHA – REAL (KIND=nag_wp) *Input*
On entry: the scalar α .
- 7: A(LDA,*) – REAL (KIND=nag_wp) array *Input*
Note: the second dimension of the array A must be at least $\max(1, K)$ if TRANS = 'N' and at least $\max(1, N)$ if TRANS = 'T'.
On entry: the matrix A ; A is n by k if TRANS = 'N', or k by n if TRANS = 'T'. If ALPHA = 0.0, A is not referenced.
- 8: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F06WCF (DSFRK) is called.
Constraints:
 if TRANS = 'N', $LDA \geq \max(1, N)$;
 if TRANS = 'T', $LDA \geq \max(1, K)$.
- 9: BETA – REAL (KIND=nag_wp) *Input*
On entry: the scalar β .
- 10: CR($N \times (N + 1)/2$) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the upper or lower triangular part (as specified by UPLO) of the n by n symmetric matrix C , stored in RFP format (as specified by TRANSR). The storage format is described in detail in Section 3.3.3 in the F07 Chapter Introduction.

On exit: the updated matrix C , that is its upper or lower triangular part stored in RFP format.

6 Error Indicators and Warnings

None.

7 Accuracy

Not applicable.

8 Parallelism and Performance

F06WCF (DSFRK) is not threaded by NAG in any implementation.

F06WCF (DSFRK) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example reads in the lower triangular part of a symmetric matrix C which it converts to RFP format. It also reads in α , β and a 6 by 4 matrix A and then performs the symmetric rank-4 update $C \leftarrow \alpha AA^T + \beta C$.

10.1 Program Text

```

Program f06wcf

!      F06WCF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dsfrk, dtfttr, dtrttf, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: alpha, beta
      Integer                    :: i, ifail, info, k, lda, n
      Character (1)              :: trans, transr, uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:,,:), c(:,,:), cr(:)
!      .. Executable Statements ..
      Write (nout,*) 'F06WCF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Read (nin,*) n, k, uplo, transr, alpha, beta, trans

      lda = n
      Allocate (c(lda,n),cr((n*(n+1))/2),a(lda,k))

```

```

!      Read upper or lower triangle of matrix C from data file
      If (uplo=='L' .Or. uplo=='l') Then
        Do i = 1, n
          Read (nin,*) c(i,1:i)
        End Do
      Else
        Do i = 1, n
          Read (nin,*) c(i,i:n)
        End Do
      End If

!      Read matrix A from data file
      Read (nin,*)(a(i,1:k),i=1,n)

!      Convert C to rectangular full packed storage in cr

!      The NAG name equivalent of dtrttf is f01vef
      Call dtrttf(transr,uplo,n,c,lda,cr,info)

      Write (nout,*)
      Flush (nout)

!      Perform the rank-k update

!      The NAG name equivalent of dsfrk is f06wcf
      Call dsfrk(transr,uplo,trans,n,k,alpha,a,lda,beta,cr)

!      Convert cr back from rectangular full packed to standard format in C

!      The NAG name equivalent of dtfttr is f01vgf
      Call dtfttr(transr,uplo,n,cr,c,lda,info)

!      Print out the result, stored in the lower triangle of matrix C

      ifail = 0
      Call x04caf('Lower','N',n,n,c,lda,'The Solution',ifail)

      End Program f06wcf

```

10.2 Program Data

F06WCF Example Program Data

```

6 4 'L' 'N' 4.21 0.89 'N'      : N, K, UPLO, TRANSR, ALPHA, BETA, TRANS
1.0
2.0 2.0
3.0 3.0 3.0
4.0 4.0 4.0 4.0
5.0 5.0 5.0 5.0 5.0
6.0 6.0 6.0 6.0 6.0 6.0 : End of matrix C
3.21 1.32 2.31 0.25
1.65 1.87 0.32 -1.54
1.80 2.88 2.05 -0.89
5.25 -2.95 -0.95 -3.80
1.58 -2.69 -2.90 -1.04
-1.11 -0.66 -0.59 0.80      : End of matrix A

```

10.3 Program Results

F06WCF Example Program Results

The Solution

	1	2	3	4	5	6
1	74.3339					
2	35.9614	38.3792				
3	61.9998	46.3791	72.2571			
4	44.8769	40.1617	13.6156	220.8276		
5	-18.4440	-2.9162	-37.3241	101.0169	85.3835	
6	-18.2242	-13.5482	-19.1635	-21.4356	9.1315	16.5209
